

Bulletin de l'EPI n° 42 de juin 1986

Émilien Pélisset

► **To cite this version:**

Émilien Pélisset. Bulletin de l'EPI n° 42 de juin 1986. EPI (Association Enseignement Public & Informatique) , 1986, ISSN : 1254-3985 ; <http://www.epi.asso.fr>. edutice-00000830

HAL Id: edutice-00000830

<https://edutice.archives-ouvertes.fr/edutice-00000830>

Submitted on 25 May 2005

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

LE NOUVEAU LECTEUR THOMSON (SUITE)

Michel CANAL

Il vous est peut-être arrivé d'aller chez un collègue avec vos disquettes LSE et de ne pas pouvoir vous en servir; votre lecteur étant un ancien modèle en simple densité, celui de votre collègue étant un nouveau modèle double densité.

Si votre collègue ne s'est pas encore procuré la disquette UTIL3 D256 (voir bulletin 40 page 128 et 129) il vous reste une solution efficace: celle de transformer votre disquette en double densité, voici comment :

Pour cela vous chargez le BASIC DISK sur le THOMSON de votre collègue, vous tapez le programme ci-dessous puis vous le lancez.

Ce programme copie la face 0 sur la face 1 initialise la face 0 en double densité recopie la face 1 sur la face 0.

Pour remettre votre disquette dans l'état initial il suffirait de taper les commandes suivantes (ce n'est utile que si vous n'avez pas de copie) :

| | |
|---------------|-------------------------------------|
| DENSITY 1,1 | sélection simple densité sur face 1 |
| DENSITY 0,1 | sélection simple densité sur face 0 |
| DSKINI 0 | réinitialisation en simple densité |
| BACKUP 1 TO 0 | restauration de votre disquette |

Notes: Ce programme n'est valable que pour des disquettes LSE L'écriture de ce programme en LSE n'est pas possible sans PBi.

```

10 REM COPIE DES SECTEURS EXISTANTS
20 REM ACCROCHAGE AUX ANCIENS
30 A$=CHR$(27):PRINT A$;CHR$(96);A$;CHR$(71);A$;CHR$(80);CHR$(12)
40 REM LA FACE 0 PASSE EN DOUBLE DENSITE
50 REM LA FACE 1 = COPIE ORGINAL
60 REM
70 CLEAR 3000:DIM S$(16),D(16)
80 PRINT "TRANSFORMATION DE DISQUETTE L.S.E."
90 PRINT

```

```

100 PRINT "PASSAGE SE SIMPLE EN DOUBLE DENSITE"
110 PRINT:PRINT:PRINT
120 PRINT "METTRE LA DISQUETTE A TRAITER"
130 PRINT "PRET";:INPUT C$
140 IF C$<>"0" THEN END
150 PRINT:PRINT "INITIALISATION FACE 1..."
160 DENSITY 1,1:DSKINI 1
170 DENSITY 0,1
180 PRINT:PRINT "COPIE FACE 0 -> FACE 1"
190 FOR P=0 TO 39:FOR S=1 TO 16
200 S$(S)=DSKI$(O,P,S):NEXT S
210 FOR S=1 TO 16:DSKO$ 1,P,S,S$(S):NEXT S
220 PRINT ".":NEXT P
230 PRINT:PRINT "REINITIALISATION FACE 0..."
240 DENSITY 0,2:DSKINI 0
250 PRINT:PRINT "TRANSFERT FACE 1 -> FACE 0"
260 FOR P=0 TO 39: FOR S=1 TO 15 STEP 2
270 REM S$= 2 SECTEURS DE 128 OCTETS
280 S$=DSKI$(1,P,S):T$=DSKI$(1,P,S+1)
290 D(S)=ASC(MID$(T$,128,1)):D(S+1)=0 300 S$(S)=S$+MID$(T$,1,127)
310 Q=P:T=S+3:IF T=18 THEN T=2:Q=Q+1
320 IF Q=40 THEN Q=0:T=0
330 IF Q=1 AND T=1 THEN Q=0:T=0
340 S$(S+1)=CHR$(Q)+CHR$(T)
350 NEXT S
360 FOR S=1 TO 16
370 X=PEEK(&H604F)*256+PEEK(&H6050)+255 380 POKE X,D(S)
390 DSKO$ 0,P,S,S$(S):NEXT S
400 PRINT ".":NEXT P
410 PRINT "MISE A JOUR CATALOGUE"
420 S$=DSKI$(0,0,3)
430 X=PEEK(&H604F)*256+PEEK(&H6050)
440 P=PEEK(X+31):S=PEEK(X+32)
450 T=PEEK(X+33)*256+PEEK(X+34)+312
460 T1=INT(T/256):T2=T-T1*256
470 T$=MID$(S$,1,31)+CHR$(39)+CHR$(16)+CHR$(T1)+CHR$(T2)
    +MID$(S$,36,220)
480 DSKO$ 0,0,3,T$
490 S$=CHR$(1)+CHR$(2)+MID$(DSKI$(0,S,P),3,253):DSKO$ 0,S,P,S$

```

```
500 S=5:P=0
510 S$=DSKI$(0,P,S)
520 Q=ASC(S$):T=ASC(MID$(S$,2,1))
530 IF Q+T<>0 THEN P=Q:S=T:GOTO 510
540 S$=CHR$(0)+CHR$(6)+MID$(S$,3,253):DSKO$ 0,P,S,S$
550 PRINT "OPERATION TERMINEE":END
```

Michel CANAL